# $Id: PKGBUILD 166698 2012-09-15 23:58:57Z eric $
# Maintainer: Eric Bélanger <eric@archlinux.org>
# Contributor: Lukas Jirkovsky <l.jirkovsky@gmail.com>

pkgname=lcms2
pkgver=2.4
pkgrel=1
pkgdesc="Small-footprint color management engine, version 2"
arch=('i686' 'x86_64')
license=('MIT')
depends=('libtiff')
url="http://www.littlecms.com"
options=('!libtool')
source=(http://downloads.sourceforge.net/sourceforge/lcms/${pkgname}-${pkgver}.tar.gz)
sha1sums=('9944902864283af49e4e21a1ca456db4e04ea7c2')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

check() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make check
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
